4 CLEAR ,,,32768!
5 REM **** SLOT ****
6 KEY OFF : GOSUB 5000
8 GOSUB 2500
10 CLS
11 REM **** DRAW BOARD ****
12 COLOR 10,8
13 DIM CHERRY(800),BAR(800),BELL(800),PLUM(800),ORANGE(800),SEVEN(800)
14 GOSUB 9999 :LOCATE 20,11:PALETTE 2,14:PRINT " H A V E    F U N  ":FOR WAITER=1 TO 1000:NEXT :CLS:PALETTE 2,12
15 LINE(0,0)-(319,20),,B
17 LINE(2,2)-(317,18),,B,&H5555
20 LINE (20,55)-(80,104),,B
25 LINE (119,104)-(181,55),,B
30 LINE (220,104)-(280,55),,B
36 LINE(0,170)-(315,191),,B
39 LINE (2,172)-(313,189),,B,&H5555
41 LINE (3,128)-(72,157),,B
42 LINE (5,130)-(70,155),,B,&H5555
48 LOCATE 2,13: PRINT "LAS VEGAS CLUB"
190 REM **** START PROGRAM ****
201 Z=R
202 W$="T225O1C16O2C16O3C16O4C16O5C16O6C16"
205 Y$="MNT120 O3 P8 O2E8 O2G8. O2E16 F8 G4.O3 P8 O2E8 O2G8. O2E16 F8 G4 "
209 RANDOMIZE TIMER
210 IF Z<=0 THEN 1000
211 IF Z=>3 THEN GOSUB 1405
212 IF Z=2 THEN GOSUB 1500
213 IF Z=1 THEN GOSUB 1600
214 IF Z=>500 THEN GOTO 8000
217 K$=INKEY$
218 IF K$="" THEN 217
219 IF K$=CHR$(57) THEN 3000
220 IF K$=CHR$(49) THEN 250
221 IF K$=CHR$(50) AND Z=>2 THEN 260
222 IF K$=CHR$(51) AND Z=>3 THEN 270
224 BEEP:GOTO 217
250 GOSUB 800
255 GOTO 300
260 GOSUB 800
265 GOTO 300
270 GOSUB 800
300 A=INT(RND*6+1)
301 B=INT(RND*6+1)
302 C=INT(RND*6+1)
402 IF A=1 THEN PUT(10,45),BAR
403 IF A=2 THEN PUT(10,45),CHERRY
404 IF A=3 THEN PUT(11,45),PLUM
405 IF A=4 THEN PUT(13,43),BELL
406 IF A=5 THEN PUT(12,45),ORANGE
407 IF A=6 THEN PUT(10,45),SEVEN
408 PLAY "V8L64T200N46"
409 IF B=1 THEN PUT(111,45),BAR
410 IF B=2 THEN PUT(111,45),CHERRY
411 IF B=3 THEN PUT(111,45),PLUM
412 IF B=4 THEN PUT(112,43),BELL
413 IF B=5 THEN PUT(113,45),ORANGE
414 IF B=6 THEN PUT(111,45),SEVEN
415 PLAY "V8L64T200N46"
416 IF C=1 THEN PUT(210,45),BAR
417 IF C=2 THEN PUT(210,45),CHERRY
418 IF C=3 THEN PUT(211,45),PLUM
419 IF C=4 THEN PUT(212,43),BELL
420 IF C=5 THEN PUT(212,45),ORANGE
421 IF C=6 THEN PUT(210,45),SEVEN
423 PLAY "V8L64T200N46"
425 IF K$= CHR$(51) THEN Z=Z-3 : GOSUB 700
426 IF K$= CHR$(50) THEN Z=Z-2 : GOSUB 600
427 IF K$= CHR$(49) THEN Z=Z-1 : GOSUB 500
428 FOR WAITER =1 TO 1300 : NEXT
429 GOSUB 475
430 FOR WAITER=1 TO 300:NEXT
431 LOCATE 18,2:PRINT "       ":LOCATE 19,3:PRINT "      "
432 LOCATE 18,2:PRINT "BALANCE"
433 LOCATE 19,3:PRINT "$"Z;
435 IF Z=<1 THEN 465
436 IF Z=<2 THEN 465
445 FOR WAITER=1 TO 1300 : NEXT
450 LOCATE 18,2:PRINT "       "
451 LOCATE 19,3:PRINT "      "
460 GOTO 209
465 FOR WAITER =1 TO 800 : NEXT
468 GOTO 209
475 LOCATE 8,4:PRINT "       ":LOCATE 9,4:PRINT "       ":LOCATE 10,4:PRINT"       ":LOCATE 11,4:PRINT "       ":LOCATE 12,4:PRINT"       ":LOCATE 13,4:PRINT"       "
476 LOCATE 8,16:PRINT "       ":LOCATE 9,16:PRINT "       ":LOCATE 10,16:PRINT "       ":LOCATE 11,16:PRINT "       ":LOCATE 12,16:PRINT "       ":LOCATE 13,16:PRINT "       "
477 LOCATE 8,29:PRINT "       ":LOCATE 9,29:PRINT "       ":LOCATE 10,29:PRINT "       ":LOCATE 11,29:PRINT "       ":LOCATE 12,29:PRINT "       ":LOCATE 13,29:PRINT "       "
480 RETURN
500 LOCATE 18,3
506 REM ***** MAKE PAYOFF FOR 1 COIN *****
510 IF A=1 AND B=1 AND C=1 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 50" :PLAY Y$
520 IF A=2 AND B=2 AND C=2 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 20": PLAY W$ :GOTO 581
530 IF A=3 AND B=3 AND C=3 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 20": PLAY W$
540 IF A=4 AND B=4 AND C=4 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 20": PLAY W$
545 IF A=5 AND B=5 AND C=5 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 20": PLAY W$
547 IF A=6 AND B=6 AND C=6 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 100" :PLAY Y$
550 IF A=3 AND B=3 AND C=1 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 18": PLAY W$
560 IF A=4 AND B=4 AND C=1 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 18": PLAY W$
565 IF A=5 AND B=5 AND C=1 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 18": PLAY W$
570 IF A=2 AND B=2 THEN LOCATE 18,3: PRINT "WINNER":LOCATE 19,3: PRINT"$ 5": PLAY W$: GOTO 581
580 IF A=2 THEN LOCATE 18,3 : PRINT "WINNER" : LOCATE 19,3:PRINT "$ 2":PLAY W$
581 IF A=1 AND B=1 AND C=1 THEN Z=Z+50
582 IF A=2 AND B=2 AND C=2 THEN Z=Z+20 : GOTO 595
583 IF A=3 AND B=3 AND C=3 THEN Z=Z+20
584 IF A=4 AND B=4 AND C=4 THEN Z=Z+20
585 IF A=5 AND B=5 AND C=5 THEN Z=Z+20
586 IF A=6 AND B=6 AND C=6 THEN Z=Z+100
587 IF A=3 AND B=3 AND C=1 THEN Z=Z+18
588 IF A=4 AND B=4 AND C=1 THEN Z=Z+18
589 IF A=5 AND B=5 AND C=1 THEN Z=Z+18
590 IF A=2 AND B=2 THEN Z=Z+5 : GOTO 595
591 IF A=2 THEN Z=Z+2
595 RETURN
600 LOCATE 18,3
606 REM ***** MAKE PAYOFF FOR 2 COINS *****
610 IF A=1 AND B=1 AND C=1 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 100" :PLAY Y$
620 IF A=2 AND B=2 AND C=2 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 40" :PLAY W$: GOTO 681
630 IF A=3 AND B=3 AND C=3 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 40" :PLAY W$
640 IF A=4 AND B=4 AND C=4 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 40" :PLAY W$
645 IF A=5 AND B=5 AND C=5 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 40" :PLAY W$
647 IF A=6 AND B=6 AND C=6 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 200" :PLAY Y$
650 IF A=3 AND B=3 AND C=1 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 36" :PLAY W$
660 IF A=4 AND B=4 AND C=1 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 36" :PLAY W$
665 IF A=5 AND B=5 AND C=1 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 36" :PLAY W$
670 IF A=2 AND B=2 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 10": PLAY W$ :GOTO 681
680 IF A=2 THEN LOCATE 18,3 : PRINT "WINNER":LOCATE 19,3:PRINT "$ 4": PLAY W$
681 IF A=1 AND B=1 AND C=1 THEN Z=Z+100
682 IF A=2 AND B=2 AND C=2 THEN Z=Z+40 : GOTO 695
683 IF A=3 AND B=3 AND C=3 THEN Z=Z+40
684 IF A=4 AND B=4 AND C=4 THEN Z=Z+40
685 IF A=5 AND B=5 AND C=5 THEN Z=Z+40
686 IF A=6 AND B=6 AND C=6 THEN Z=Z+200
687 IF A=3 AND B=3 AND C=1 THEN Z=Z+36
688 IF A=4 AND B=4 AND C=1 THEN Z=Z+36
689 IF A=5 AND B=5 AND C=1 THEN Z=Z+36
690 IF A=2 AND B=2 THEN Z=Z+10: GOTO 695
691 IF A=2 THEN Z=Z+4
695 RETURN
700 LOCATE 18,3
706 REM ***** MAKE PAYOFF FOR 3 COINS *****
710 IF A=1 AND B=1 AND C=1 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 150" :PLAY Y$
720 IF A=2 AND B=2 AND C=2 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 60":PLAY W$ :GOTO 781
730 IF A=3 AND B=3 AND C=3 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 60":PLAY W$
740 IF A=4 AND B=4 AND C=4 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 60":PLAY W$
745 IF A=5 AND B=5 AND C=5 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 60":PLAY W$
747 IF A=6 AND B=6 AND C=6 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 300" :PLAY Y$
750 IF A=3 AND B=3 AND C=1 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 54":PLAY W$
760 IF A=4 AND B=4 AND C=1 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 54":PLAY W$
765 IF A=5 AND B=5 AND C=1 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 54":PLAY W$
770 IF A=2 AND B=2 THEN PRINT "WINNER":LOCATE 19,3:PRINT "$ 20":PLAY W$: GOTO 781
780 IF A=2 THEN LOCATE 18,3: PRINT "WINNER":LOCATE 19,3:PRINT "$ 6" :PLAY W$
781 IF A=1 AND B=1 AND C=1 THEN Z=Z+150
782 IF A=2 AND B=2 AND C=2 THEN Z=Z+60 : GOTO 795
783 IF A=3 AND B=3 AND C=3 THEN Z=Z+60
784 IF A=4 AND B=4 AND C=4 THEN Z=Z+60
785 IF A=5 AND B=5 AND C=5 THEN Z=Z+60
786 IF A=6 AND B=6 AND C=6 THEN Z=Z+300
787 IF A=3 AND B=3 AND C=1 THEN Z=Z+54
788 IF A=4 AND B=4 AND C=1 THEN Z=Z+54
789 IF A=5 AND B=5 AND C=1 THEN Z=Z+54
790 IF A=2 AND B=2 THEN Z=Z+20: GOTO 795
791 IF A=2 THEN Z=Z+6
795 RETURN
799 REM **** NUMBER OF COINS PLAYED ****
800 LOCATE 23,7 :PRINT "                               "
810 IF K$=CHR$(49) THEN LOCATE 23,14:PRINT "1 COIN PLAYED"
815 IF K$=CHR$(50) THEN LOCATE 23,14:PRINT "2 COIN'S PLAYED"
820 IF K$=CHR$(51) THEN LOCATE 23,14:PRINT "3 COIN'S PLAYED"
830 RETURN
1000 CLS
1002 REM ***** GAME OVER  *****
1003 COLOR 0:PALETTE 2,14
1005 LINE (0,0)-(319,180),,B
1006 LINE (4,4)-(315,176),,B,&H5555
1009 IF Z<R THEN R=R-Z
1010 LOCATE 8,13 : PRINT "GAME OVER ";Q$
1019 LOCATE 20,3 : PRINT "YOU LOST $";R "WANT TO TRY AGAIN Y/N ?"
1022 PLAY "T120 MN MF O3"
1023 PLAY "C8.C16.F2.C8.F16.A2.C8.F16 A4 C8.F16 A4 C8.F16 A2."
1024 PLAY "F8.A16 ML> C2 MN< A4 F4 C2.C8.C16 ML F1 MN F4"
1026 P$=INKEY$
1027 IF P$="" THEN 1026
1028 IF P$=CHR$(89) OR P$=CHR$(121) THEN CLEAR : GOTO 8
1029 IF P$=CHR$(78) OR P$= CHR$(110) THEN 1040
1035 BEEP : GOTO 1026
1040 SYSTEM
1405 REM **** 3 OR MORE COINS ****
1408 LOCATE 23,7 : PRINT "                            "
1410 LOCATE 23,7 : PRINT "ENTER 1 TO 3 COIN'S PLEASE"
1450 RETURN
1500 REM **** ONLY 2 COINS ****
1505 LOCATE 18,2:PRINT "       "
1507 LOCATE 19,3:PRINT "      "
1510 LOCATE 23,7 : PRINT "                          "
1513 LOCATE 23,7 : PRINT "ENTER UPTO 2 COIN'S PLEASE"
1550 RETURN
1600 REM **** ONLY 1 COIN ****
1610 LOCATE 18,2:PRINT "       "
1611 LOCATE 19,3:PRINT "      "
1612 LOCATE 23,7: PRINT "                             "
1613 LOCATE 23,7: PRINT " ENTER 1 COIN ONLY PLEASE"
1650 RETURN
2500 CLS : SCREEN 5: COLOR 10,8:PALETTE 2,12
2501 LOCATE 2
2502 PRINT SPC(9) "____________________" : PRINT
2503 PRINT SPC(5) ">>> D O L L A R  S L O T <<<"
2504 PRINT SPC(9) "____________________"
2508 LOCATE 8,8
2510 PRINT "This is a Dollar Slot, you start with "
2511 PRINT  "upto $25 and can play from 1 to 3 coins.
2512 PRINT : PRINT "Your bet will be substracted from your  balance."
2515 PRINT: PRINT : INPUT "Type in your name please";Q$
2520 LOCATE 18 :INPUT "Enter amount needed";R
2531 IF R=<25 THEN 2544
2533 LOCATE 20,5 : IF R>25 THEN PRINT" You can start with upto $25 try again."
2535 BEEP
2536 FOR WAITER=1 TO 1500 : NEXT
2537 LOCATE 20,5 : PRINT "                                                "
2538 LOCATE 18,22: PRINT "   "
2539 GOTO 2520
2544 LOCATE 23,5 : PRINT "Remember to cash out ENTER 9 "
2545 FOR WAITER =1 TO 2500 : NEXT
2550 CLS:RETURN
3000 CLS
3001 IF Z<R THEN 1000
3003 COLOR 0: PALETTE 2,9
3005 REM ***** CASHING OUT *****
3010 LOCATE 5
3011 FOR I= 1 TO 10 : PRINT TAB(8) "YOU ";Q$ " HAVE WON $";Z-R
3012 NEXT I
3013 LOCATE 5
3014 FOR I= 1 TO 9:  PRINT TAB(8) "                        "
3015 NEXT I
3020 LINE (0,0)-(319,180),,B
3021 LINE (4,4)-(315,176),,B,&H5555
3024 PLAY "T140 MN MF"
3026 PLAY "O2 G4.A4 B8 > C4.<B4 A8 G4 A8 G4 E8 D4.C4."
3030 PLAY "O2 G4.A4 B8 > C4.<B4 A8 G4 A8 G4 E8 D4.C4."
3045 FOR I=1 TO 5000:NEXT I
3051 CLS
3055 SYSTEM
5000 CLS:SCREEN 3:KEY OFF
5001 REM **** DRAW TITLE ****
5010 C=4
5110 FOR L=100 TO 0 STEP -5
5120 L1=L1+5:LINE(L1,0)-(0,L),C
5130 IF C=4 THEN C=1 ELSE C=4
5140 LINE(159-L1,0)-(159,L),C
5150 NEXT
5160 L1=0
5170 FOR L=100 TO 199 STEP 5
5180 L1=L1+5:LINE(L1,199)-(0,L),C
5190 IF C=4 THEN C=1 ELSE C=4
5200 LINE(159-L1,199)-(159,L),C
5205 NEXT
5209 COLOR 14
5210 LINE (0,0)-(159,199),1,B
5211 LOCATE 7,5 : PRINT "DOLLAR SLOT"
5212 LOCATE 11,10:PRINT "BY"
5215 LOCATE 15,5 : PRINT "TONY GERVASI"
5240 FOR WAITER= 1 TO 4000:NEXT
5500 RETURN
8000 REM **** BROKE THE BANK ****
8001 CLS
8003 COLOR 0: PALETTE 2,14
8020 LINE (0,0)-(319,180),,B
8025 LINE (4,4)-(315,176),,B,&H5555
8050 FOR F=1 TO 5 : SOUND 500,1 : SOUND 200,1 : SOUND 150,1 : NEXT
8060 LOCATE 10,10:PRINT "YOU BROKE THE BANK !!!!"
8070 FOR F=1 TO 10 :SOUND 500,1:SOUND 200,1:SOUND 150,1:NEXT
8072 LINE (57,138)-(263,170),,B
8073 LINE (59,140)-(261,168),,B,&H5555
8075 LOCATE 20,9:PRINT "WANT TO PLAY AGAIN Y/N ?"
8100 P$=INKEY$
8110 IF P$=CHR$(89) OR P$=CHR$(121) THEN CLEAR : GOTO 8
8120 IF P$=CHR$(78) OR P$= CHR$(110) THEN SYSTEM
8130 GOTO 8100
9000 REM *** DRAW-CHERRY-PLUM-BAR-ORANGE-SEVEN-BELL ***
9999 LOCATE 3,18:PRINT "CHERRY"
10000 CIRCLE(65,54),9,2,0,6.2831
10100 PAINT(63,53),2,2
10110 CIRCLE(84,48),9,2,0,6.2831
10120 PAINT(84,48),2,2
10130 LINE(80,26)-(84,42),1:LINE(80,26)-(65,48),1
10140 GET(35,10)-(115,80),CHERRY:CLS
10145 LOCATE 3,19:PRINT "BELL"
10150 DRAW "C3BM64,34R18M90,60L35M64,34":PAINT(89,59),3,3
10160 CIRCLE(73,62),4,3,0,6.2831
10170 PAINT(76,62),3,3
10180 LINE(69,61)-(77,61),0
10190 CIRCLE(73,36),9,3,0,6.2831
10200 PAINT(73,30),3,3
10210 GET(35,10)-(115,80),BELL:CLS
10215 LOCATE 3,19:PRINT "BAR"
10220 LINE(54,27)-(96,34),3,B
10230 PAINT(55,28),3,3
10240 LINE(54,55)-(96,62),3,B
10250 PAINT(55,57),3,3
10260 LINE(54,41)-(96,48),3,B
10270 PAINT(55,42),3,3
10280 GET(35,10)-(115,80),BAR:CLS
10285 LOCATE 3,19:PRINT "PLUM"
10300 CIRCLE(73,45),20,1,0,6.2831,1/1.8
10310 PAINT(75,45),1,1
10320 DRAW "C1BM93,44R2D1L2D1R2"
10330 GET(35,10)-(115,80),PLUM:CLS
10335 LOCATE 3,19:PRINT "ORANGE"
10340 CIRCLE(73,45),20,2,0,6.2831
10350 PAINT(75,45),2,2
10360 DRAW "C1BM71,27R4D1L4R1D1R2"
10370 GET(35,10)-(115,80),ORANGE:CLS
10375 LOCATE 3,19:PRINT "SEVEN"
10380 LINE(61,30)-(90,30),1
10390 LINE(90,30)-(77,61),1
10400 LINE(77,61)-(68,61),1
10410 LINE(68,61)-(77,38),1
10420 LINE(77,38)-(61,38),1
10430 LINE(61,38)-(61,30),1
10440 PAINT(63,33),1,1
10450 GET(35,10)-(115,80),SEVEN:CLS
10460 RETURN
